

Práctica: Sistema de Identificación por Radiofrecuencia (RFID) con Arduino

Practice: Radio Frequency Identification (RFID) System with Arduino

Iván H. Pérez-Tavera ^a

Abstract:

It is important for high school students to develop computational thinking and skills in the STEM (Science, Technology, Engineering, and Mathematics) fields. Arduino is presented as one of the most accessible tools to foster these competencies through laboratory practices. Based on the above, this document presents a practical guide for the design, assembly, and programming of a Radio Frequency Identification (RFID) system using the Arduino microcontroller and the RC522 reader module. During the activity, the student will learn to program the reading of unique identifiers (UID) from passive tags to manage an access control system, integrating visual (LEDs and LCD screen) and mechanical (servo motor) outputs to indicate whether access is granted or denied.

Keywords:

Arduino, RFID, RC522, Access control, Programming, STEM

Resumen:

Es importante que los alumnos de bachillerato desarrollen el pensamiento computacional y habilidades en el área STEM (Ciencia, Tecnología, Ingeniería y Matemáticas). Arduino se presenta como una de las herramientas más accesibles para fomentar estas competencias a través de prácticas de laboratorio. Con base en lo anterior, este documento presenta una guía práctica para el diseño, ensamblaje y programación de un sistema de Identificación por Radiofrecuencia (RFID) utilizando el microcontrolador Arduino y el módulo lector RC522. Durante la actividad, el estudiante aprenderá a programar la lectura de identificadores únicos (UID) de etiquetas pasivas para gestionar un sistema de control de acceso, integrando salidas visuales (LEDs y pantalla LCD) y mecánicas (servomotor) para indicar la concesión o denegación del acceso.

Palabras Clave:

Arduino, RFID, RC522, control de acceso, programación, STEM

Introducción

La Identificación por Radiofrecuencia (RFID) es una tecnología que permite identificar y rastrear objetos de forma inalámbrica mediante el uso de ondas de radio. Es un sistema similar al de los códigos de barras, pero con la ventaja de que no requiere una línea de visión directa para la lectura.

El sistema se compone de dos elementos fundamentales:

1. **El Lector:** Es el dispositivo encargado de emitir una señal de radiofrecuencia y recibir la respuesta de las etiquetas. En esta práctica,

utilizaremos el módulo RFID-RC522, un lector de bajo costo muy popular para proyectos con microcontroladores.

2. **La Etiqueta (Tag):** Es un dispositivo pequeño que contiene una antena y un microchip con un identificador único (UID) y, en algunos casos, datos adicionales. Las etiquetas pueden ser pasivas (no tienen batería propia y se activan con la energía que reciben del lector) o activas (con batería propia, lo que les da mayor alcance). Usaremos etiquetas pasivas en forma de tarjeta y llavero.

^a Iván Horacio Pérez Tavera, Universidad Autónoma del Estado de Hidalgo | Escuela Preparatoria No. 4 | Pachuca-Hidalgo | México,
<https://orcid.org/0000-0001-7708-6912>, Email: itavera@uaeh.edu.mx

¿Cómo funciona?

Cuando acercamos una etiqueta pasiva al lector, la antena del lector crea un campo electromagnético. La antena de la etiqueta capta este campo y, mediante inducción, alimenta su chip interno. El chip envía su UID de vuelta al lector, que luego transmite esta información a nuestro Arduino a través de un protocolo de comunicación llamado SPI (Serial Peripheral Interface). El Arduino compara el UID recibido con una base de datos interna y decide qué acción realizar.

Aplicaciones comunes:

- Control de accesos (tarjetas de identificación escolar, llaves de hotel).
- Inventarios y logística (seguimiento de productos).
- Tarjetas de transporte público (como la tarjeta del Metro de la CDMX).
- Identificación de mascotas (microchips).

Funcionamiento del control de acceso

Materiales

- ✓ Microcontrolador Arduino Uno.
- ✓ Módulo Lector RFID-RC522.
- ✓ Pantalla LCD con módulo I2C
- ✓ Micro Servo motor SG90
- ✓ Etiqueta RFID pasiva (en forma de tarjeta).
- ✓ Etiqueta RFID pasiva (en forma de llavero).
- ✓ Cables Jumper (macho-hembra y macho-macho).
- ✓ Protoboard.
- ✓ 2 LED (rojo y verde)
- ✓ 2 resistencias de 220Ω

Se realiza la conexión de los componentes con la guía del siguiente esquema:




Ilustración 1 Esquema de conexión (creación propia en WOKWI)

Esquema de conexiones del sistema RFID con Arduino

Pin RFID	Pin Arduino	Función	Notas
GND	GND	Tierra	Conexión a tierra común.
3.3V	3.3V	Alimentación	El módulo funciona a 3.3V No conectarlo a 5V
RST	Pin 9	Reset	Pin para reiniciar el módulo.
SDA (SS)	Pin 10	Serial Data (Slave Select)	Pin de selección del dispositivo SPI.
MOSI	Pin 11	Master Out Slave In	Transmisión de datos del Arduino al lector.
MISO	Pin 12	Master In Slave Out	Transmisión de datos del lector al Arduino.
SCK	Pin 13	Serial Clock	Señal de reloj de la comunicación SPI.
IRQ	No conectado	Interrupt Request	No se usa en esta práctica básica.

Se propone el siguiente código para la realización de la práctica:

```
1 #include <LiquidCrystal_I2C.h>
2 #include <Servo.h>
3 #include <SPI.h>
4 #include <MFRC522.h>
5
6 // =====
7 // 1. CONFIGURACIÓN DE CONSTANTES Y PINES
8 // =====
9
10 // UID Autorizados
11 const String UID_AUTORIZADO = " 11 22 33 44";
12 const String UID_NO_AUTORIZADO = " 33 3e 8b 04";
13
14 // Pines para el módulo MFRC522
15 #define SS_PIN 10 // Pin Slave Select (SDA/SS)
16 #define RST_PIN 9 // Pin Reset
17
18 // Pines para LEDs
19 int LedRojo = 5;
20 int LedVerde = 3;
21
22 // Objetos
23 MFRC522 mfrc522(SS_PIN, RST_PIN);
24 Servo servoMotor;
25
26 LiquidCrystal_I2C lcd(0x27, 16, 2);
27
28 // --- Funciones para organización de código---
29 void reset_system();
30 void handle_access_granted();
31 void handle_access_denied();
32 String read_current_uid();
```

```

34 // =====
35 // 2. SETUP
36 // =====
37
38 void setup() {
39   // Inicialización de dispositivos
40   lcd.init(); //Inicializa LCD
41   lcd.backlight();
42   lcd.clear();
43   lcd.begin(16, 2);
44   SPI.begin(); // Inicializar SPI
45   mfrc522.PCD_Init(); // Inicializar el módulo MFRC522
46   servoMotor.attach(2);
47   pinMode(LedRojo, OUTPUT);
48   pinMode(LedVerde, OUTPUT);
49
50   // Estado inicial del sistema
51   reset_system();
52 }
53
54 // =====
55 // 3. LOOP
56 // =====
57
58 void loop() {
59   // 1. Verificar si hay una tarjeta RFID cerca del lector
60   if (mfrc522.PICC_IsNewCardPresent() && mfrc522.PICC_ReadCardSerial()) {
61
62     // 2. Leer y formatear el UID
63     String current_uid = read_current_uid();
64
65     // 3. Analizar acceso
66     if (current_uid == UID_AUTORIZADO) {
67       handle_access_granted();
68     } else {
69       handle_access_denied();
70     }
71
72     // 4. Detener la lectura del PICC
73     mfrc522.PICC_HaltA();
74     mfrc522.PCD_StopCrypto1();
75
76     // 5. REINICIAR: Espera y vuelve a la pantalla de solicitud
77     delay(500); // Pausa breve antes de limpiar la pantalla
78     reset_system();
79   }
80
81   // El sistema solo espera aquí hasta que se detecte una tarjeta.
82 }

```

```

84 // =====
85 // 4. FUNCIONES
86 // =====
87
88 /**
89  * Lee el UID del sensor y lo formatea la cadena usada para la comparación.
90  */
91 String read_current_uid() {
92   String uid = "";
93   // Mostrar mensaje de lectura
94   lcd.clear();
95   lcd.setCursor(0, 0);
96   lcd.print("Leyendo tarjeta...");
97   lcd.setCursor(0, 1);
98   lcd.print("Identificando...");
99
100   for (byte i = 0; i < mfrc522.uid.size; i++) {
101     // Añade el espacio y el byte en formato hexadecimal (HEX)
102     uid += String(mfrc522.uid.uidByte[i] < 0x10 ? "0" : " ");
103     uid += String(mfrc522.uid.uidByte[i], HEX);
104   }
105   uid.toLowerCase();
106   return uid;
107 }
108
109 /**
110  * Muestra Acceso Concedido, activa el servo/LED verde y espera 3 segundos.
111  */
112 void handle_access_granted() {
113   lcd.clear();
114   lcd.setCursor(0, 0);
115   lcd.print("ACCESO CONCEDIDO!");
116   lcd.setCursor(0, 1);
117   lcd.print("Puerta Abierta");
118
119   servoMotor.write(60); // Abre la puerta
120   digitalWrite(LedVerde, HIGH); // LED Verde ON
121   digitalWrite(LedRojo, LOW); // LED Rojo OFF
122
123   delay(3000); // Mantiene la acción por 3 segundos
124 }

```

```

126 /**
127  * Muestra Acceso Denegado, activa el LED rojo y espera 3 segundos.
128  */
129 void handle_access_denied() {
130   lcd.clear();
131   lcd.setCursor(0, 0);
132   lcd.print("ACCESO DENEGADO");
133   lcd.setCursor(0, 1);
134   lcd.print("UID No Valido");
135
136   servoMotor.write(0); // Mantiene el servo cerrado
137   digitalWrite(LedRojo, HIGH); // LED Rojo ON
138   digitalWrite(LedVerde, LOW); // LED Verde OFF
139
140   delay(3000); // Mantiene el mensaje de error por 3 segundos
141 }
142
143 /**
144  * Resetea el estado físico (Servo, LEDs) y la pantalla LCD
145  * para solicitar un nuevo escaneo.
146  */
147 void reset_system() {
148   // 1. Reseteo dispositivos (Servo a 0, LEDs apagados)
149   servoMotor.write(0);
150   digitalWrite(LedRojo, LOW);
151   digitalWrite(LedVerde, LOW);
152
153   // 2. Limpiar y establecer la pantalla LCD con el mensaje de solicitud
154   lcd.clear();
155   lcd.setCursor(0, 0);
156   lcd.print("Sist. Seguridad");
157   lcd.setCursor(0, 1);
158   lcd.print("Acercar Tarjeta...");
159 }

```

En las siguientes imágenes se muestra el funcionamiento dentro del simulador de wokwi:




Ilustración 2 Inicialización del sistema (creación propia en WOKWI)

Se hace la simulación de acercar una tarjeta no permitida, por lo que el sistema niega el acceso:




Ilustración 3 Acceso denegado (creación propia en WOKWI)

Se muestra la simulación de la tarjeta de acceso correcta, lo cual se indica por pantalla y el servo hace el movimiento indicando que la puerta se abrió.




Ilustración 4 Acceso Concedido (creación propia en WOKWI)

Conclusión

Esta práctica demuestra la viabilidad de integrar tecnologías de identificación por radiofrecuencia (RFID) con microcontroladores Arduino para el diseño de sistemas automatizados de control de acceso. El ensamblaje y la programación del módulo lector RC522, en sincronía con actuadores mecánicos (servomotor) y visuales (pantalla LCD y LEDs), permite al estudiante materializar conceptos de lógica computacional y protocolos de comunicación (SPI) en interacciones físicas reales. Si bien la lectura estática de un Identificador Único (UID) proporciona una introducción funcional a la arquitectura de hardware abierto, también expone de manera evidente las limitaciones de seguridad y vulnerabilidades a la clonación inherentes a los sistemas de autenticación de nivel básico. Por lo tanto, este ejercicio no solo sienta las bases técnicas de la mecatrónica y la automatización, sino que obliga a desarrollar un criterio analítico sobre la fiabilidad de los métodos de seguridad en entornos reales.

Referencias

- [1] Del Valle Hernández, L. (2022, 14 septiembre). Lector RFID RC522 control de acceso RFID con Arduino. Programarfacil Arduino y Home Assistant. <https://programarfacil.com/blog/arduino-blog/lector-rfid-rc522-con-arduino/>.
- [2] BUONO UNO R3 with RC522 RFID Reader Kit for Makers. (2019, 08 mayo). Arduino Project Hub. <https://projecthub.arduino.cc/ZJH/buono-uno-r3-with-rc522-rfid-reader-kit-for-makers-ee12fb>.